




普通高等教育“十二五”规划教材



普通高等教育“十一五”国家级规划教材配套教材



汇编语言设计及 接口与通信实践指导书

庞春江 牛为华 编著



中国电力出版社
CHINA ELECTRIC POWER PRESS



普通高等教育“十二五”规划教材



普通高等教育“十一五”国家级规划教材配套教材

汇编语言设计及 接口与通信实践指导书

编著 庞春江 牛为华
主审 滕桂法



中国电力出版社
CHINA ELECTRIC POWER PRESS

内 容 提 要

本书是普通高等教育“十二五”规划教材、普通高等教育“十一五”国家级规划教材配套教材，根据接口与通信技术的课程要求，结合近年来实验教学改革的实践经验，针对提高学生的计算机应用开发和创新能力的教学目标而编写。

本书内容包括汇编语言及硬件课程实验基础，综合设计实验和创新型实验设计四部分。重点突出设计与应用创新，创新性实验以问题和课题为核心、以学生为主体进行设计，注重激发学生的创新思维，提高创新实践能力。全书既注重计算机硬件与软件知识相结合，又注重操作性与设计性。

本书可作为高等院校计算机类及电气类、电子类相关专业本科学生的实验教材，也可供有关技术人员参考。

图书在版编目（CIP）数据

汇编语言设计及接口与通信实践指导书 / 庞春江, 牛为华编著. —北京: 中国电力出版社, 2011.5

普通高等教育“十二五”规划教材 普通高等教育“十一五”国家级规划教材配套教材

ISBN 978-7-5123-1637-9

I. ①汇… II. ①庞… ②牛… III. ①汇编语言—程序设计—高等学校—教材②微型计算机—接口技术—高等学校—教材③计算机通信—高等学校—教材 IV. ①TP313
②TP364.7③TN919

中国版本图书馆 CIP 数据核字（2011）第 075216 号

中国电力出版社出版、发行

（北京市东城区北京站西街 19 号 100005 <http://www.cepp.sgcc.com.cn>）

航远印刷有限公司印刷

各地新华书店经售

*

2011 年 5 月第一版 2011 年 5 月北京第一次印刷

787 毫米×1092 毫米 16 开本 7.5 印张 179 千字

定价 13.00 元

敬 告 读 者

本书封面贴有防伪标签，加热后中心图案消失

本书如有印装质量问题，我社发行部负责退换

版 权 专 有 翻 印 必 究

前 言

目前，我国正在加快开发具有自主知识产权的计算机软、硬件产品，并进行大规模生产技术改造，因此，对既具备软件编程能力又了解硬件知识的复合型人才的需求和要求在不断提高。

接口与通信技术作为计算机类及大多数电类专业学生的必修课程，是操作性、设计性都很强的、计算机硬件与软件相结合的课程。为方便学生掌握这门课程并具备一定的设计和编程能力，在计算机应用开发和创新方面打下良好的基础，作者在以素质教育和创新能力培养为目标的教学理念下，在总结实验教学改革实践和经验的基础上编写了该实验指导书。

该书的编写注重内容更新，重点突出设计与应用创新，即综合运用所学的基础知识，设计与编制具有较复杂功能的应用系统，培养工程意识和创新能力。为此我们将实践环节分为三个阶段来实施。

第一阶段为基础性与认知性阶段：主要安排了以验证性为主的软件和硬件功能模块实验，这部分内容基本按照理论教学进度安排。其中包括：程序结构实验，8255 并行接口实验，基于 8253 的定时/计数器应用实验，基于 8259 的中断实验，基于 ADC0809 的 A/D 实验以及多个功能器件的综合实验。初学者可将更多的注意力集中于对接口电路功能的理解、编程方法及接口电路工作中软、硬件的配合。

第二阶段为综合运用与设计性提高阶段：综合性实践总结近几年实践性教学环节的成果，旨在培养学生综合运用所学知识解决实际问题的能力。在选题和实现方式上力求做到：①与理论课教学内容相配合，并在实现技术和芯片运用知识方面作适当拓展。②各部分设计题目之间既有一定的独立性又相互配合。

第三阶段为创新型设计阶段：该部分实践内容是以教育部“大学生创新性实验计划”为宗旨，以问题和课题为核心，即联系实际又结合学生的实践能力来设计的。以本科学生为主体的创新性实验设计，可以激发学生的创新思维和创新意识，逐渐掌握思考问题、解决问题的方法，提高其创新实践能力。

本书是与《接口与通信》（王保义编著）配套的实验指导书，在编写过程中得到了河北农业大学滕桂法教授的大力支持，滕教授在百忙中审阅了全书，郑顾平老师对本书的编写提出了宝贵意见，在此表示诚挚的谢意。另外，接口实验采用 LAB6000 改进型伟福实验装置设计者设计的部分实验，在此向该装置开发者致谢。

由于编者的经验和水平所限，书中难免会有不足和错漏之处，恳请本行同仁及读者指正。

编 者
2011 年 2 月

目 录

前言

| | |
|------------------------|----|
| 1 汇编语言课程实验基础篇 | 1 |
| 实验一 数码转换程序设计及调试 | 1 |
| 实验二 数据运算类程序设计 | 3 |
| 实验三 分支及循环程序设计 | 4 |
| 实验四 排序程序设计 | 6 |
| 实验五 系统中断调用 | 8 |
| 实验六 判断闰年程序设计 | 10 |
| 实验七 播放乐曲程序设计 | 12 |
| 实验八 图形显示程序设计 | 16 |
| 实验九 磁盘读写访问实验 | 20 |
| 2 硬件课程实验基础篇 | 24 |
| 实验十 存储器扩展实验 | 24 |
| 实验十一 8255 并行接口实验 | 26 |
| 实验十二 中断特性及 8259 应用编程实验 | 29 |
| 实验十三 8253 定时/计数应用实验 | 36 |
| 实验十四 模/数 (A/D) 转换实验 | 41 |
| 实验十五 数/模 (D/A) 转换实验 | 44 |
| 实验十六 键盘扫描与八段数码管显示实验 | 47 |
| 实验十七 点阵 LED 显示实验 | 51 |
| 实验十八 液晶显示控制实验 | 56 |
| 3 综合性实验应用提高篇 | 65 |
| 实验十九 交通信号灯控制实验 | 65 |
| 实验二十 双机通信实验 | 66 |
| 实验二十一 温度传感器综合应用实验 | 69 |
| 实验二十二 直流电动机控制综合实验 | 70 |
| 4 创新型设计实践篇 | 72 |
| 实验二十三 计算器设计实验 | 72 |
| 实验二十四 计分设计实验 | 74 |
| 实验二十五 高级动画设计实验 | 74 |
| 实验二十六 电压表设计实验 | 76 |
| 实验二十七 急救车与交通灯设计实验 | 77 |
| 实验二十八 高级电子时钟及日历设计实验 | 82 |

| | |
|----------------------------|------------|
| 实验二十九 智能风扇设计实验 | 83 |
| 实验三十 转速表设计实验 | 87 |
| 实验三十一 智能洗衣机设计实验 | 88 |
| 实验三十二 出租车计价器设计实验 | 93 |
| 附录 | 95 |
| 附录 A 汇编语言程序设计规范及上机过程 | 95 |
| 附录 B 实验总体要求与方法 | 98 |
| 附录 C WAFE6000 仿真系统介绍 | 99 |
| 附录 D WAFE 系统组成与结构图 | 100 |
| 附录 E 伟福系统软件开发环境 | 107 |
| 附录 F 程序调试 DEBUG 主要命令 | 110 |
| 附录 G 8086/8088 常用指令表 | 111 |
| 参考文献 | 114 |

1 汇编语言课程实验基础篇

实验一 数码转换程序设计及调试

一、实验目的

- (1) 学习并掌握在 PC 机上建立、编制、连接和运行 8086/8088 汇编语言程序。
- (2) 对实验程序进行调试，掌握不同进制数转换程序的编制和调试。
- (3) 了解简单的程序设计方法和程序调试手段。

二、实验类型

验证性实验。

三、实验内容

1. 二进制到 BCD 码转换

计算机中的数值有各种表达方式，掌握各种数制之间的转换是一种基本功。本实验要求将给定的一个二进制数转换成十进制（BCD）码。二进制到 BCD 码转换程序框图如图 1-1 所示。实验流程和参考程序如下。

```
; 将 AX 拆为三个 BCD 码，并存入 Result 开始的三个单元
DATA SEGMENT
Result DB 3 DUP(?)
DATA ENDS
CODE SEGMENT
ASSUME CS:CODE, DS: DATA
START PROC NEAR
    MOV AX, DATA
    MOV DS, AX
    MOV AX, 123
    MOV CL, 100
    DIV CL
    MOV Result, AL      ; 除以 100, 得百位数
    MOV AL, AH
    MOV AH, 0
    MOV CL, 10
    DIV CL
    MOV Result+1, AL    ; 余数除以 10, 得十位数
    MOV Result+2, AH    ; 余数为个位数
    JMP $
CODE ENDS
END START
```




图 1-1 二进制到 BCD 码转换程序框图

2. 二进制到 ASCII 码转换

此实验主要让学生了解数值的 BCD 码和 ASCII 码的区别，利用查表功能可快速地进行数值转换。将给出的 BCD 数转换成 ASCII 值。二进制到 ASCII 码转换程序框图如图 1-2 所示。

实验流程和参考程序如下。

```
; 将 AX 拆为两个 ASCII 码，并存入 Result 开始的两个单元
DATA SEGMENT
Result DB 2 DUP(?)
ASCIIITab:
    DB "0123456789ABCDEF" ; 定义数字对应的 ASCII 表
DATA ENDS
CODE SEGMENT
ASSUME CS:CODE, DS: DATA
START PROC NEAR
MOV AX, DATA
MOV DS, AX
MOV AX, 1AH
MOV BL, AL
SHR BL, 4
MOV BH, 0
MOV AH, BYTE PTR ASCIIITab[BX]
MOV Result, AH
AND AL, 0FH
MOV BL, AL
MOV AH, BYTE PTR ASCIIITab[BX]
MOV Result+1, AH
JMP $
CODE ENDS
END START
```

3. 十六进制到 ASCII 码转换

十六进制数加 30H 即可得到 0~9H 的 ASCII 码，十六进制数加 37H 即可得到 A~FH 的 ASCII 码。本实验要求将任意给定的一个十六进制数转换成 ASCII 码。

假设二字节十六进制数存放在 3800H 的内存单位中，把它们转换为 ASCII 码后，分别存入到开始地址为 380AH 的四个单元中。本程序稍复杂些，使用到转移指令，入栈/出栈指令。

实验参考程序如下。

```
CODE SEGMENT
ASSUME CS:CODE
START: MOV CX, 0004H
       MOV DI, 3800H
       MOV DX, [DI]
L0:  MOV AX, DX
       AND AX, 000FH
       CMP AL, 0AH
       JB LA
       ADD AL, 07H
LA:  MOV [DI], AL
       ADD DI, 2H
       LOOP L0
```



图 1-2 二进制到 ASCII 码
转换程序框图

```

LA: ADD AL, 30H
    MOV [DI+0AH], AL
    INC DI
    PUSH CX
    MOV AL, 04H
    SHR DX, CL
    POP CX
    LOOP L0
    HLT
CODE ENDS
END START

```

实验二 数据运算类程序设计

一、实验目的

- (1) 学习并掌握在 PC 机上建立、编制、连接和运行 8086/8088 汇编语言程序。
- (2) 对实验程序进行调试，掌握不同的进制数转换程序的编制和调试。
- (3) 掌握简单运算和分支及循环程序设计

方法。

二、实验类型

验证性实验。

三、实验内容

将两个十进制数相乘，并在屏幕上显示结果。已知乘数和被乘数均以 BCD 码形式存放在 DATA1 和 DATA2 为首的单元中（低位位于低地址），被乘数为单字节数，结果送回 RESULT 处。

数据运算程序设计实验流程图如图 1-3 所示，参考程序如下。

```

DATA SEGMENT
DATA1 DB 5 DUP (?)
DATA2 DB ?
RESULT DB 6 DUP (?)
DATA ENDS
CODE SEGMENT
ASSUME CS: CODE, DS: DATA
START: MOV AX, DATA
        MOV DS, AX
        MOV SI, OFFSET DATA2
        MOV BL, [SI]
        AND BL, 0FH
        CMP BL, 09H
        JNC ERROR
        MOV SI, OFFSET DATA1
        MOV DI, OFFSET RESULT
        MOV CX, 0005H

```




图 1-3 数据运算程序设计实验流程图

```

A1:    MOV AL, [SI+04H]
        AND AL, 0FH
        CMP AL, 09H
        JNC ERROR
        DEC SI
        MUL BL
        AAM
        ADD AL, [DI+05H]
        AAA
        MOV [DI+05H], AL
        DEC DI
        MOV [DI+05H], AH
        LOOP A1
        MOV CX, 06H
        MOV SI, OFFSET RESULT
DISPLAY: MOV AH, 02H
        MOV AL, [SI]
        ADD AL, 30H
        MOV DL, AL
        INT 21H
        LOOP DISPLAY
        JMP A2
ERROR:  MOV DL, 45H
        MOV AH, 02H
        INT 21H
A2:    MOV AH, 4CH
        INT 21H
CODE   ENDS
END START

```

实验三 分支及循环程序设计

一、实验目的

- (1) 进一步学习并掌握在 PC 机上建立、编制、连接和运行调试汇编语言程序。
- (2) 掌握分支结构程序和循环结构程序设计方法。

二、实验类型

验证性实验。

三、实验内容

实验前复习教材中有关分支结构和循环结构程序设计方法的内容。

1. 求符号数最大、最小值

设在内存缓冲区 BUF+1 单元开始，存放 8 个单字节数，其数据长度在 BUF 单元，要求找出最大数送 MAX 单元，最小数送 MIN 单元。

假设第一个数是最大数，并送到 MAX 单元中，依次把余下的数与 MAX 单元中的数相比较，如大于 MAX 中的数，则把此数送到 MAX 单元中，否则取下一个数，最后 MAX 单元即为最大数。同理可求出最小数。

求符号数最大、最小值实验流程图如图 1-4 所示。

完成以下程序的调试并验证结果，注意分支判断条件和循环结构。通过单步调试，细致观察数据比较后在存储单元内的变化。

```

DATA SEGMENT
BUF DB 8,22,33,12,-1,-50,-32,97,127
MAX DB ?
MIN DB ?
DATA ENDS
CODE SEGMENT
ASSUME CS: CODE,
DS: DATA
BEG: MOV AX, DATA
MOV DS, AX
MOV CH, 0
MOV CL, BUF
DEC CX
MOV AL, BUF+1
MOV MAX, AL
MOV MIN, AL
MOV BX, OFFSET BUF+2
LAST: MOV AL, [BX]
CMP AL, MAX
JG GREAT
CMP AL, MIN
JL LESS
JMP NEXT
GREAT:MOV MAX, AL
JMP NEXT
LESS: MOV MIN, AL
NEXT: INC BX
LOOP LAST
MOV AH, 4CH
INT 21H
CODE ENDS
END BEG

```

2. 计算数列中负数的个数

假设数据长度存放在内存地址为 3000H 的单元中，从 3001H 单元开始连续存放若干个数据，要求计算这些数中负数的个数并保存在数据区的最后一个单元中。

计算数列中负数的个数实验流程图如图 1-5 所示，参考程序如下。

```

CODE SEGMENT
ASSUME CS: CODE
START: MOV DI, 3000H
MOV CL, [DI]
XOR CH, CH
MOV BL, CH
INC DI
A1: MOV AL, [DI]
TEST AL, 80H
JE A2
INC BL
A2: INC DI
LOOP A1

```

```

MOV [DI], BL
MOV AH, 4CH
INT 21H
CODE    ENDS
END START

```

完成以上程序的调试并验证结果。




图 1-4 求符号数最大值、最小值实验流程图




图 1-5 计算数列中负数的个数实验流程图

实验四 排序程序设计

一、实验目的

- (1) 进一步学习并掌握在 PC 机上建立、编制、连接和运行调试汇编语言程序。
- (2) 重点掌握分支结构、循环结构、子程序结构程序设计方法。

二、实验类型

综合性实验。

三、实验内容

实验前复习教材中有关分支结构程序和循环结构程序的内容，复习排序、查找算法。编写程序实现学生成绩的排序。

将分数为 1~100 之间的 30 个成绩存入首址为 3001H 的单元中，3000H+I 表示学号为 I 的学生成绩。编写程序在 3100H 开始的区域排出名次表，3100H+I 为学号为 I 的学生名次。

设计思路如下。

(1) 先把学生成绩的最高分即第一名找出来，并记录下学号，根据学号送入相应的名次表单元中。

(2) 然后查找第二名，第三名……，直至最后一名。

(3) 这样只要报出学号就可以知道该学生的成绩排名了。把查找最高分的程序作为子程序，查找的次数作为外循环。

完成相应程序的编制，进行调试并验证结果。

排序程序设计实验流程图如图 1-6 所示。

其子程序流程图如图 1-7 所示。图中：CX，(循环计数) 入栈保护，返回时注意出栈恢复；CX，(循环计数) 学生数，便于成绩比较；AL，存放最低成绩 00H；SI，指向成绩表首址；DX，记录对应学号。




图 1-6 排序程序设计实验流程图

CX—学生人数，循环计数；DI—指向排序名次表首址；

SI—指向成绩表首址




图 1-7 子程序流程图

实验五 系统中断调用

一、实验目的

- (1) 掌握在 PC 机上建立、编写、连接和运行调试汇编语言程序。
- (2) 掌握 DOS 系统中断调用程序设计方法，实现人机对话。

二、实验类型

验证性实验。

三、实验内容

1. 人机会话程序

通过人机会话可以获取程序运行所需要的数据，本程序应用的 DOS 中断功能，涉及 1、2、9、0AH 四个功能的调用。

设计思路：首先计算机询问用户姓名，等待用户输入姓名的汉语拼音，然后将它显示在屏幕上，请用户认可，用户键入“Y”程序结束，否则再次询问用户姓名。

人机会话程序实验流程图如图 1-8 所示，参考程序如下。

```

DATA      SEGMENT
MESG1    DB  0DH,0AH,'What is your name ? $'
MESG2    DB  '? (Y/N) $'
BUF       DB   30
          DB  ?
          DB  30 DUP (?)
DATA      ENDS
CODE      SEGMENT
          ASSUME CS: CODE, DS: DATA
BEG:      MOV  AX, DATA
          MOV  DS, AX
AGAIN:    MOV  AH, 9
          MOV  DX, OFFSET MESG1
          INT  21H
          MOV  AH, 0AH
          MOV  DX, OFFSET BUF
          INT  21H
          MOV  BL, BUF+1
          MOV  BH, 0
          MOV  SI, OFFSET BUF+2
          MOV  BYTE PTR [BX+SI], '$'
          MOV  AH, 2
          MOV  DL, 0AH
          INT  21H
          MOV  AH, 9
          MOV  DX, OFFSET BUF+2
          INT  21H
          MOV  AH, 9
MON     DB  OFFSET MESG2

```

```

INT 21H
MOV AH, 1
INT 21H
CMP AL, 'Y'
JNZ AGAIN
MOV AH, 4CH
INT 21H
CODE ENDS
END BEG

```

2. 接收小写字母转换成大写字母并显示

从键盘上接收一串小写字母存入内存单元，经过转换后将其在屏幕上显示为大写字母。

设计思路：利用 DOS 中断的 01 号子功能接收键盘输入的一串字符串并存入内存单元，当接收到回车键时，将小写字母的 ASCII 码转换为大写字母的 ASCII 码，利用 DOS 功能的 02 号子功能将其显示在屏幕上。

大小写字母转换程序实验流程图如图 1-9 所示，参考程序如下。




图 1-8 人机会话程序实验流程图




图 1-9 大小写字母转换程序实验流程图

```

DATA SEGMENT
BUF DB 200 DUP (?)
DATA ENDS
CODE SEGMENT
ASSUME CS: CODE, DS: DATA
START: MOV AX, DATA
       MOV DS, AX

```

```

        MOV SI,OFFSET BUF
        MOV CL,00H
A1:    MOV AH,01H
        INT 21H
        CMP AL,0DH
        JZ A2
        SUB AL,20H
        MOV[SI],AL
        INC SI
        INC CL
        JMP A1
A2:    MOV DL,0AH
        MOV AH,02H
        INT 21H
        MOV DL,0DH
        INT 21H
        MOV SI,OFFSET BUF
        MOV CH,00H
A3:    MOV DL,[SI]
        MOV AH,02H
        INT 21H
        INC SI
        LOOP A3
        MOV AH,4CH
        INT 21H
CODE   SEGMENT
        END START

```

3. 日期显示

接收键盘输入的年、月、日，并在屏幕上显示。

设计思路：先在屏幕上显示“WHAT IS THE DATE？(YYYY/MM/DD)”，然后接收键盘输入的年、月、日信息，再依次在屏幕上显示。

完成程序的编制，进行调试并验证结果。

日期显示实验流程图如图 1-10 所示。




图 1-10 日期显示实验流程图

实验六 判断闰年程序设计

一、实验目的

- (1) 掌握在 PC 机上建立、编写、连接和运行调试汇编语言程序。
- (2) 通过调试该程序，进一步掌握键盘接收数据并进行运算的方法。

二、实验类型

验证性实验。

三、实验内容

通过调试该程序，进一步掌握键盘接收数据并进行运算的方法。年份从键盘输入，通过

运行计算后，判断某年是否为闰年，输出相应信息。

设计思路：利用 DOS 21H 终端类型的 10 号键盘功能调用，将从键盘接收的年份数值型字符串按字节存入缓冲区。将字符串中的每个字符转换为对应的十进制数值（利用 0~9 这 10 个数的 ASCII 码值比其本身大 30H 的关系），通过判断某年是否为闰年的表达式，判断输出是否为闰年的信息。

判断某年是否为闰年的表达式为

$((AX) \text{ MOD } 4 \text{ EQ } 0) \text{ AND } ((AX) \text{ MOD } 100 \text{ NE } 0) \text{ OR } ((AX) \text{ MOD } 400 \text{ EQ } 0)$
其中，(AX) 中存放的是从键盘输入的年份值。

该表达式的意思是：如果某一年能被 4 整除但不能被 100 整除或者能被 400 整除，则该年为闰年。

自行设计“判断闰年子程序”，与主程序一起调试，并验证结果。例如：从键盘输入“2003”，则从屏幕输出 “This is not a leap year”。判断闰年子程序实验流程图如图 1-11 所示，参考程序如下。

```

DATA SEGMENT
INFON DB 0DH,0AH,'PLEASE INPUT A YEAR: $'
Y     DB 0DH,0AH,'THIS IS A LEAP YEAR! $'
N     DB 0DH,0AH,'THIS IS NOT A LEAP YEAR! $'
W     DW 0
BUF   DB 8
      DB ?
      DB 200 DUP (?)
DATA  ENDS
STACK SEGMENT STACK
DB 200 DUP (?)
STACK ENDS
CODE  SEGMENT
ASSUME CS: CODE,
DS: DATA, SS: STACK
START: MOV AX, DATA
       MOV DS, AX
       LEA DX, INFON
       MOV AH, 09H
       INT 21H
       LEA DX, BUF
       MOV AH, 0AH
       INT 21H
       MOV CL, [BUF+1]
       LEA DI, BUF+2
       CALL DATACATE
       CALL IFYEARS
       JC A1
       LEA DX, NMOV AH, 09H
       INT 21H
       JMP EXIT
A1:   LEA DX, Y
       MOV AH, 09H

```




图 1-11 判断闰年子程序实验流程图